

Vim to T_EXShop Preview Scripts

Nicola Vitacolonna

This note documents the procedure that enables two-way synchronisation between Vim (MacVim) and T_EXShop. The setup works for L^AT_EX documents as well as ConT_EXt documents.

1 Backward (PDF to Source) Synchronisation

Three steps are needed:

1. Check Preferences > Source > Configure for External Editor.
2. To sync from T_EXShop preview to MacVim, create a file `othereditor` in `/usr/local/bin` with the following content:

```
#!/bin/sh
/usr/local/bin/mvim --remote-silent +"$1" "$2"
```

This will allow T_EXShop to jump to line `$1` of file `$2` in MacVim when you cmd-click somewhere in the preview. A similar command can be used with regular Vim, provided that it has been compiled with `+clientserver` (which is the default for MacVim).

3. Make the file executable:

```
chmod 755 /usr/local/bin/othereditor
```

2 Forward (Source to PDF) Synchronisation

To jump from a L^AT_EX or ConT_EXt source opened in (Mac)Vim to the corresponding PDF preview in TeXShop, you need three Vim scripts. The first defines two functions, one to open a PDF in T_EXShop, and another that calls an AppleScript script to jump to the point in the PDF that corresponds to a given line and column

in the source. The latter has two additional parameters to configure the synchronisation method for \LaTeX or ConTeXt , respectively. This script should be saved in `~/.vim/autoload/tex.vim`. This is the content of the script:

```
fun! tex#preview()
  call job_start(['open', '-a', 'TeXShop.app', expand("%:p")])
endf

fun! tex#goto_texshop(path, linenr, colnr, context = 0, altpath = 0)
  call job_start(['osascript',
    \ '-e', 'tell front document of application "TeXShop"',
    \ '-e', 'activate',
    \ '-e', 'SyncWithOtherEditor',
    \ '-e', a:altpath ? 'AlternateBinPath' : 'StandardBinPath',
    \ '-e', a:context ? 'SyncConTeXt' : 'SyncRegular',
    \ '-e', printf('sync_preview_line theLine %d', a:linenr),
    \ '-e', printf('sync_preview_index theIndex %d', a:colnr),
    \ '-e', printf('sync_preview_name theName "%s"', a:path),
    \ '-e', 'end tell',
    \ ])
endf
```

The other two scripts define buffer-local mappings (which you may change as you like) to open a PDF in \TeX Shop and to jump to a given position in the PDF. For \LaTeX save these mappings in `~/.vim/after/ftplugin/tex.vim`:

```
nnoremap <silent><buffer> <leader>tv :<c-u>call tex#preview()<cr>
nnoremap <silent><buffer> <leader>ts
  \ :<c-u>call tex#goto_texshop(expand("%:p"), line('.'), col('.'), 0, 0)<cr>
```

For ConTeXt , save these mappings in `~/.vim/after/ftplugin/context.vim`:

```
nnoremap <silent><buffer> <leader>tv :<c-u>call tex#preview()<cr>
nnoremap <silent><buffer> <leader>ts
  \ :<c-u>call tex#goto_texshop(expand("%:p"), line('.'), col('.'), 1, 1)<cr>
```

With the setup above, you may type `\tv` (where `\` is Vim's leader) to open and view in \TeX Shop the PDF document corresponding to the `.tex` file in the current buffer. You may type `\ts` to jump to the position in the PDF corresponding to the current cursor's position in Vim. The latter mapping assumes that *the PDF file has already been opened in \TeX Shop (with `\tv`), and it is the front document*.

3 Further Configuration

It is recommended that you add some “magic lines” to your source documents. In particular, you should specify the engine used to typeset your document:

```
% !TEX TS-program = <engine>
```

This is because, when T_EXShop is configured to use an external editor, it will typeset your document the first time it is opened. For L^AT_EX documents, that should be all you need. Read further if you use ConT_EXt.

At the time of this writing (July 2021), the latest ConT_EXt beta is needed for synchronisation to work (the version in T_EX Live 2021 will not work!). To work with the latest beta proceed as follows:

- Download and install ConT_EXt as explained in [ConT_EXt Garden](#).
- In T_EXShop, set Preferences > Engine > Alternate Path to the path of your ConT_EXt beta installation.
- In your source documents, add the following magic lines:

```
% !TEX useAlternatePath  
% !TEX useConTeXtSyncParser
```

- Make sure that SyncT_EX is enabled in your source documents:

```
\setupsynctex[state=start,method=min]
```

4 External Resources

- [Using T_EXShop with ConT_EXt](#)
- [Using ConT_EXt in Vim](#)